如何使用Ardino UNO連接和驅動4位7段LED顯示屏
這很簡單，但它需要很多電線。通常我們需要顯示由多個數字組成的數字。為此，可以在市場上找到具有共陽極或共陰極的4位7段LED顯示器。
該模塊包含四個7段LED數字顯示器。這些數字可以獨立打開和關閉。模塊中的四個數字中的每一個使用其自己的共陰極（或共陽極）連接點。這意味著每個段（線）共享相同的陽極（或陰極）連接點。段標記為：A，B，C，D，E，F，G，H（對於點）。每個段都連接到所有四個數字。這減少了所需的佈線（4 + 8 = 12對4×8 = 32）。

有12個連接點= 4個PIN用於數字+ 8個PIN用於段（線）。

Segment G( pin6) 錯誤 應改為pin5
1位和4位顯示器之間最重要的區別是我們打開和關閉段的方式。為了點亮多個數字，我們使用'多路復用'。

我們通過無限循環快速循環顯示4位數字。

1）通過切換連接到共陰極（陽極）的PIN來打開有效數字。

2）確定要打開的段列表並點亮這些段。

3）等待一段時間（2-5毫秒）。

4）關閉數字（陰極和所有段）。

6）選擇列表[1 ... 4]中的下一個數字。

Arduino UNO R3規範稱每個PIN的最大電流為40 mA。所以我們需要一些限制電阻。要點亮LED，我們需要大約15 mA。電阻值可以計算為：

R = V / I，

V = 5伏

I = 10..20 mA

R = 220 ... 470歐姆

當所有段都打開時，有些人意識到電路中有太多的電流。電流為7 * 20mA = 140 mA，或7 * 10 mA = 70 mA（取決於電阻值）。這種電流對Arduino是安全的，但會損壞STM32 MCU（晶體管可以解決這個問題）。

現在是時候將SMA420564 4位7段LED顯示屏連接到Arduino UNO。 Arduino UNO Basic Learning Kit中提供了所需的組件。我們需要4個電阻器，麵包板和12個電線。將限制220歐姆的電阻連接到4位LED顯示屏的12,9,8,6 PIN。應根據此表連接電線。下為測試程式源:
nt segA = 3; // top
int segB = 4; // right-top
int segC = 5; // right-bottom
int segD = 6; // bottom
int segE = 7; // left-bottom
int segF = 8; // left-top
int segG = 9; // middle
 

int digit1 = 14; // common cathode for digit1 (i.e. A0)
int digit2 = 15; // common cathode for digit2 (i.e. A1)
int digit3 = 16; // common cathode for digit3 (i.e. A2)
int digit4 = 17; // common cathode for digit4 (i.e. A3)
 

// Number to display
static int number = 0;
static int ms = 0;
 

//==============================================================//
void setup()
{
    // All pins in digital mode
    pinMode(digit1, OUTPUT);
    pinMode(digit2, OUTPUT);
    pinMode(digit3, OUTPUT);
    pinMode(digit4, OUTPUT);
 

    pinMode(segA, OUTPUT);
    pinMode(segB, OUTPUT);
    pinMode(segC, OUTPUT);
    pinMode(segD, OUTPUT);
    pinMode(segE, OUTPUT);
    pinMode(segF, OUTPUT);
    pinMode(segG, OUTPUT);
}
 

//==============================================================//
void loop() {
    // put your main code here, to run repeatedly:
 

    if( millis() / 1000 > ms )
    {
        ms = millis() / 1000;
        if( ++number > 9999 ) number = 0;
    }
 

    // activate common cathode for digit1
    digitalWrite(digit1, LOW);
        // light on segments for 'thousands'
        drawDigitFast( (number/1000) % 10 );
        // wait 2 ms
        delay(2);
        // turn off all segments
        drawDigitFast( -1 );
    // turn off cathode for digit1
    digitalWrite(digit1, HIGH);
 

    digitalWrite(digit2, LOW);
        drawDigitFast( (number/100) % 10 );
        delay(2);
        drawDigitFast( -1 );
    digitalWrite(digit2, HIGH);
 

    digitalWrite(digit3, LOW);
        drawDigitFast( (number/10) % 10 );
        delay(2);
        drawDigitFast( -1 );
    digitalWrite(digit3, HIGH);
 

    digitalWrite(digit4, LOW);
        drawDigitFast( number % 10 );
        delay(2);
        drawDigitFast( -1 );
    digitalWrite(digit4, HIGH);
}
 

//-----------------------------------------------------//
void drawDigitFast(int n)
{
    const byte aPins[8] = {
        segA, segB, segC, segD, segE, segF, segG
    };
    const byte aSegments[11][8] = {
        //  A     B     C     D     E     F     G
        { HIGH, HIGH, HIGH, HIGH, HIGH, HIGH,  LOW }, // 0
        {  LOW, HIGH, HIGH,  LOW,  LOW,  LOW,  LOW }, // 1
        { HIGH, HIGH,  LOW, HIGH, HIGH,  LOW, HIGH }, // 2
        { HIGH, HIGH, HIGH, HIGH,  LOW,  LOW, HIGH }, // 3
        {  LOW, HIGH, HIGH,  LOW,  LOW, HIGH, HIGH }, // 4
        { HIGH,  LOW, HIGH, HIGH,  LOW, HIGH, HIGH }, // 5
        { HIGH,  LOW, HIGH, HIGH, HIGH, HIGH, HIGH }, // 6
        { HIGH, HIGH, HIGH,  LOW,  LOW,  LOW, LOW  }, // 7
        { HIGH, HIGH, HIGH, HIGH, HIGH, HIGH, HIGH }, // 8
        { HIGH, HIGH, HIGH, HIGH,  LOW, HIGH, HIGH }, // 9
        {  LOW,  LOW,  LOW,  LOW,  LOW,  LOW,  LOW }  // all off
    };
 

    if( n < 0 || n > 10 )
    {
        n = 10;
    }
 

    for( int i = 0; i < 7; i++ )
    {
        digitalWrite( aPins[i], aSegments[n][i] );
    }
}
